

## Casio fx-7700G

### Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of  $a$ ,  $b$ , and  $c$ .

```
QUADRAT
"AX2+BX+C=0"
"A="?→A
"B="?→B
"C="?→C
B2-4AC→D
D<0⇒Goto 1
"X=":(-B+√D)÷(2A)▲
"OR X=":(-B-√D)÷(2A)
Goto 2
Lbl 1
"NO REAL SOLUTION"
Lbl 2
```

### Graph Reflection Program

This program will graph a function  $f$  and its reflection in the line  $y = x$ . To use this program, enter the function in  $f_1$ .

```
REFLECT
"GRAPH -A TO A"
"A="?→A
Range -A,A,1,-2A÷3,2A÷3,1
Graph Y=f1
-A→B
Lbl 1
B→X
Plot f1,B
B+A÷32→B
B≤A⇒Goto1 :Graph Y=X
```

### Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
SOLVE
"AX+BY=C"
"A="?→A
"B="?→B
"C="?→C
"DX+EY=F"
"D="?→D
"E="?→E
"F="?→F
AE-DB=0⇒Goto 1
"X=":(CE-BF)÷(AE-DB)▲
"Y=":(AF-CD)÷(AE-DB)
Goto 2
Lbl 1
"NO UNIQUE SOLUTION"
Lbl 2
```

### Visualizing Row Operations Program not available

### Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, enter an expression in  $f_1$ .

```
EVALUATE
Lbl 1
"X="?→X
"F(X)=" : f1▲
Goto 1
```

## Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
ADDVECT
Cls
"A="?→A
"B="?→B
"C="?→C
"D="?→D
Plot 0,0
Plot A,B
Line
Plot 0,0
Plot C,D
Line ▲
A+C→E
B+D→F
Plot 0,0
Plot E,F
Line
Plot A,B
Plot E,F
Line
Plot C,D
Plot E,F
Line ▲
```

## Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing **EXE**. Press Mode Shift X to change to parametric mode when starting to write this program.

```
SINESHOW
Rad
Range -2.25,π÷2,3,-1.19,1.19,10,6.3,.15
Graph(X,Y)=(-1.25+cos T,sinT)
Graph(X,Y)=(T÷4,sinT)
0→N
Lbl 1
N+1→N
Nπ÷6.5→T
-1.25+cos T→A
sin T→B
T÷4→C
Plot A,B
Plot C,B
Line ▲
N<12⇒Goto 1
```

## Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle.

```
VECANGL
Cls
Deg
"ENTER (A,B)"
"A="?→A
"B="?→B
"ENTER (C,D)"
"C="?→C
"D="?→D
Plot 0,0
Plot A,B
Line
Plot 0,0
Plot C,D
Line ▲
AC+BD→E
√(A²+B²)→U
√(C²+D²)→V
cos⁻¹(E÷UV)→θ
"θ="
θ
```